
1 Corrections required

1.1 f_NBIOT_508CheckRRCConnectionResumeOnNewCell
	Template name
	f_NBIOT_508CheckRRCConnectionResumeOnNewCell

	Reason for change
	In the current TTCN implementation, for TCs 22.4.15 and 22.4.16, a DRB is established on NCell 11. 

However, in the function f_NBIOT_508CheckRRCConnectionResumeOn NewCell (f_NBIOT_RRC_ConnectionRelease (f_NBIOT_RRC_ConnectionRelease_Common ( f_NBIOT_RRC_ConnectionRelease_Local ( f_NBIOT_SS_SRBs_DRBs_Release, this DRB gets released as part of the local end release of SRBs and DRB.
This happens at Step 9 for testcase 22.4.15 and Step 18 for testcase 22.4.16 respectively.

After this, the postamble is invoked for USER_PLANE and the local end again tries to release an already released DRB. This happens in both 22.4.15 and 22.4.16. 

This is incorrect

Hence, it is recommended to release the DRB only once either in function f_NBIOT_508CheckRRCConnectionResumeOn NewCell or in the postamble.

	Summary of change
	Modified TTCN code by introducing a flag to ensure the DRB on NCell 11 is released only once.

	TTCN module
	NBIOT_NASSteps.ttcn

	MCC160 Comment
	Not needed


Before Change:

	function f_NBIOT_508CheckRRCConnectionResumeOnNewCell(NBIOT_CellId_Type p_CellId,
                                                        template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,
                                                        template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,
                                                        template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?) runs on NBIOT_PTC
  {
    var NAS_UL_Message_Type v_NasMsg;
    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);
    var NAS_KsiValue v_KsiValue := f_NBIOT_SecurityKSIasme_Get();
    var template (present) NAS_UL_Message_Type v_ExpectedTAU_Request;
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;
    var template (omit) AdditionalUpdateType v_AdditionalUpdateType;  /* @sic R5s170515 change 5: template (omit) sic@ */
    v_ExpectedTAU_Request := cr_508_TAU_Request(tsc_EpsUpdate_TaUpdate,
                                                v_KsiValue,
                                                cr_DRXparameter_Any ('5C'O) ifpresent,
                                                cr_AdditionalUpdateType_CIOT ifpresent);
    /* steps 1 - 3 */
    v_NasMsg := valueof(f_NBIOT_RRCConnectionResume(p_CellId, p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause, v_ExpectedTAU_Request));
    v_EPS_ContextStatus := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    v_AdditionalUpdateType := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType;
    /* steps 4 - 5 */
    f_NBIOT_TrackingAreaUpdate_Accept_Complete(p_CellId, USER_PLANE, tsc_SRB1, v_GutiParams, v_EPS_ContextStatus, v_AdditionalUpdateType);
    /* Step 6: Release RRC connection */
    f_NBIOT_RRC_ConnectionRelease(p_CellId, USER_PLANE, tsc_RRC_TI_Def);
  } 


After Change:

	function f_NBIOT_508CheckRRCConnectionResumeOnNewCell(NBIOT_CellId_Type p_CellId,
                                                        template (present) EUTRA_ASN1_ResumeIdentity_r13_Type p_ResumeIdentity := ?,
                                                        template (present) EUTRA_ASN1_ShortMAC_I_Type p_ShortMAC_I := ?,
                                                        template (present) EstablishmentCause_NB_r13 p_ResumeCause := ?) runs on NBIOT_PTC
  {
    var NAS_UL_Message_Type v_NasMsg;
    var GutiParameters_Type v_GutiParams := f_NBIOT_CellInfo_GetGuti(p_CellId);
    var NAS_KsiValue v_KsiValue := f_NBIOT_SecurityKSIasme_Get();
    var template (present) NAS_UL_Message_Type v_ExpectedTAU_Request;
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;
    var template (omit) AdditionalUpdateType v_AdditionalUpdateType;  /* @sic R5s170515 change 5: template (omit) sic@ */
    v_ExpectedTAU_Request := cr_508_TAU_Request(tsc_EpsUpdate_TaUpdate,
                                                v_KsiValue,
                                                cr_DRXparameter_Any ('5C'O) ifpresent,
                                                cr_AdditionalUpdateType_CIOT ifpresent);
    /* steps 1 - 3 */
    v_NasMsg := valueof(f_NBIOT_RRCConnectionResume(p_CellId, p_ResumeIdentity, p_ShortMAC_I, p_ResumeCause, v_ExpectedTAU_Request));
    v_EPS_ContextStatus := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    v_AdditionalUpdateType := v_NasMsg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType;
    /* steps 4 - 5 */
    f_NBIOT_TrackingAreaUpdate_Accept_Complete(p_CellId, USER_PLANE, tsc_SRB1, v_GutiParams, v_EPS_ContextStatus, v_AdditionalUpdateType);
    /* Step 6: Release RRC connection */
    f_NBIOT_RRC_ConnectionRelease(p_CellId, USER_PLANE, tsc_RRC_TI_Def,-,-,-,false); //WA#22_4_15
  }


1.2 f_NBIOT_RRC_ConnectionRelease
	Template name
	f_NBIOT_RRC_ConnectionRelease

	Reason for change
	As a result of change 1.1, this function also needs to be modified to accommodate the flag usage specified in change 1.1


	Summary of change
	Modified TTCN code by introducing a flag to ensure the DRB on NCell 11 is released only once.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	Not needed


Before Change:

	function f_NBIOT_RRC_ConnectionRelease(NBIOT_CellId_Type p_CellId,
                                         IOT_STATE_Type p_IOT_State := CONTROL_PLANE,
                                         EUTRA_ASN1_RRC_TransactionIdentifier_Type p_RRC_TI := tsc_RRC_TI_Def,
                                         ReleaseCause_NB_r13 p_ReleaseCause := other,
                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                         template (omit) NB_SRB_Identity_Type p_SrbId := omit) runs on NBIOT_PTC
  { /* @sic R5s170250: p_IOT_State instead of p_SrbId sic@ */
    f_NBIOT_RRC_ConnectionRelease_Common(p_CellId, p_IOT_State, cs_RRCConnectionRelease_NB(p_RRC_TI, p_ReleaseCause), p_TimingInfo, p_SrbId); // @sic R5s180217 sic@
  }
 


After Change:

	function f_NBIOT_RRC_ConnectionRelease(NBIOT_CellId_Type p_CellId,
                                         IOT_STATE_Type p_IOT_State := CONTROL_PLANE,
                                         EUTRA_ASN1_RRC_TransactionIdentifier_Type p_RRC_TI := tsc_RRC_TI_Def,
                                         ReleaseCause_NB_r13 p_ReleaseCause := other,
                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                         template (omit) NB_SRB_Identity_Type p_SrbId := omit,
                                         boolean p_isDrbConfigured := true /*WA#22_4_15*/) runs on NBIOT_PTC
  { /* @sic R5s170250: p_IOT_State instead of p_SrbId sic@ */
    f_NBIOT_RRC_ConnectionRelease_Common(p_CellId, p_IOT_State, cs_RRCConnectionRelease_NB(p_RRC_TI, p_ReleaseCause), p_TimingInfo, p_SrbId, p_isDrbConfigured); // WA#22_4_15 @sic R5s180217 sic@
  }



1.3 f_NBIOT_RRC_ConnectionRelease_Common
	Template name
	f_NBIOT_RRC_ConnectionRelease_Common

	Reason for change
	As a result of change 1.2, this function also needs to be modified to accommodate the flag usage specified in change 1.1.


	Summary of change
	Modified TTCN code by introducing a flag to ensure the DRB on NCell 11 is released only once.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	Not needed


Before Change:

	function f_NBIOT_RRC_ConnectionRelease_Common(NBIOT_CellId_Type p_CellId,
                                                IOT_STATE_Type p_IOT_State,
                                                template (value) DL_DCCH_Message_NB p_RRCConnectionRelease,
                                                template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                                template (omit) NB_SRB_Identity_Type p_SrbId := omit) runs on NBIOT_PTC
  { /* @sic R5s170250: p_IOT_State instead of p_SrbId sic@ */
    var NB_SRB_Identity_Type v_SrbId;
    var SubFrameTiming_Type v_TimingAtT;   // time T: sending of RRCConnectionRelease acc. to 36.523-3 clause 7A.7
    var SubFrameTiming_Type v_TimingNow := f_NBIOT_GetCurrentTiming(p_CellId);
    var integer v_MilliSecondsDelay := 300;
    var integer v_Duration;
    timer t_Timer;
    if (isvalue(p_SrbId)) {  // @sic R5s180217 sic@
     v_SrbId := valueof(p_SrbId);
    } else {
     v_SrbId := f_NBIOT_GetSrb1Id(p_IOT_State);
    }
    if (f_TimingInfo_IsNow(p_TimingInfo)) {
      v_TimingAtT := f_NBIOT_GetNextSearchSpace(p_CellId, v_MilliSecondsDelay);
    }
    else {
      // dedicated timing info is given
      v_TimingAtT := valueof(p_TimingInfo.SubFrame);
    }
    v_Duration := f_SubFrameTiming_Duration(v_TimingNow, v_TimingAtT) + tsc_NBIOT_DelayAfterRRCConnectionRelease;   /* @sic R5s170263: tsc_NBIOT_DelayAfterRRCConnectionRelease sic@ */
    t_Timer.start(int2float(v_Duration) / 1000.0);
    SRB.send(cas_NB_SRB_RrcPdu_REQ(p_CellId,
                                   v_SrbId,
                                   cs_TimingInfo_EUTRA_NB(v_TimingAtT),
                                   p_RRCConnectionRelease));
    f_NBIOT_RRC_ConnectionRelease_Local(p_CellId, p_IOT_State, v_TimingAtT);
    alt {
      [] t_Timer.timeout {}
      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any, ?)))) {   /* @sic R5s170366: ? as InitialUE_Identity sic@ */
        // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a subsequent function
        t_Timer.stop;
      }
    }
  }
 


After Change:

	function f_NBIOT_RRC_ConnectionRelease_Common(NBIOT_CellId_Type p_CellId,
                                                IOT_STATE_Type p_IOT_State,
                                                template (value) DL_DCCH_Message_NB p_RRCConnectionRelease,
                                                template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                                template (omit) NB_SRB_Identity_Type p_SrbId := omit,
                                                boolean p_isDrbConfigured := true /*WA#22_4_15*/) runs on NBIOT_PTC
  { /* @sic R5s170250: p_IOT_State instead of p_SrbId sic@ */
    var NB_SRB_Identity_Type v_SrbId;
    var SubFrameTiming_Type v_TimingAtT;   // time T: sending of RRCConnectionRelease acc. to 36.523-3 clause 7A.7
    var SubFrameTiming_Type v_TimingNow := f_NBIOT_GetCurrentTiming(p_CellId);
    var integer v_MilliSecondsDelay := 300;
    var integer v_Duration;
    timer t_Timer;
    if (isvalue(p_SrbId)) {  // @sic R5s180217 sic@
     v_SrbId := valueof(p_SrbId);
    } else {
     v_SrbId := f_NBIOT_GetSrb1Id(p_IOT_State);
    }
    if (f_TimingInfo_IsNow(p_TimingInfo)) {
      v_TimingAtT := f_NBIOT_GetNextSearchSpace(p_CellId, v_MilliSecondsDelay);
    }
    else {
      // dedicated timing info is given
      v_TimingAtT := valueof(p_TimingInfo.SubFrame);
    }
    v_Duration := f_SubFrameTiming_Duration(v_TimingNow, v_TimingAtT) + tsc_NBIOT_DelayAfterRRCConnectionRelease;   /* @sic R5s170263: tsc_NBIOT_DelayAfterRRCConnectionRelease sic@ */
    t_Timer.start(int2float(v_Duration) / 1000.0);
    SRB.send(cas_NB_SRB_RrcPdu_REQ(p_CellId,
                                   v_SrbId,
                                   cs_TimingInfo_EUTRA_NB(v_TimingAtT),
                                   p_RRCConnectionRelease));
    f_NBIOT_RRC_ConnectionRelease_Local(p_CellId, p_IOT_State, v_TimingAtT, p_isDrbConfigured); //WA#22_4_15
    alt {
      [] t_Timer.timeout {}
      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any, ?)))) {   /* @sic R5s170366: ? as InitialUE_Identity sic@ */
        // there is an RRCConnectionRequest sent by the UE which can be received by TTCN-3 in a subsequent function
        t_Timer.stop;
      }
    }
  }



1.4 f_NBIOT_RRC_ConnectionRelease_Local
	Template name
	f_NBIOT_RRC_ConnectionRelease_Common

	Reason for change
	As a result of change 1.3, this function also needs to be modified to accommodate the flag usage specified in change 1.1.


	Summary of change
	Modified TTCN code by introducing a flag to ensure the DRB on NCell 11 is released only once.

	TTCN module
	NBIOT_RRCSteps.ttcn

	MCC160 Comment
	Accepted in principle (see below)
NOTE: the proposed implementation follows the same approach as in EUTRA


Before Change:

	function f_NBIOT_RRC_ConnectionRelease_Local(NBIOT_CellId_Type p_CellId,
                                               IOT_STATE_Type p_IOT_State,
                                               SubFrameTiming_Type p_TimingAtT) runs on NBIOT_PTC
  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs (and default DRB)
     *       => this will in fact happen in the future */
    var SubFrameTiming_Type v_Timing := p_TimingAtT;
    var template (value) TimingInfo_Type v_TimingInfo;
    /* Step 1 at T: disable the UL grant */
    f_NBIOT_StopULGrantTransmission(p_CellId);
    /* Step 2 at T+64ms */
    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 64);
    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);
    if (p_IOT_State == USER_PLANE) {
      /* release security */
      f_NBIOT_SS_RRC_ReleaseSecurity(p_CellId, v_TimingInfo);
    }
    f_NBIOT_ULGrantTransmission(p_CellId, cs_NB_UL_GrantScheduling_Start(cs_NB_PeriodicGrant(-, 4)), v_TimingInfo);  /* @sic R5-177097: sic@
                                                                                                                        give 4 sub-sequent UL grants to allow the UE to do RLC ACK */
    /* Step 3 at T+80ms: Release SRBs and DRBs */
    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 80);
    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);
    f_NBIOT_SS_SRBs_DRBs_Release(p_CellId, p_IOT_State, v_TimingInfo);
    /* Step 4 at T+96ms: Configure SRBs and DRBs */
    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 96);
    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);
    f_NBIOT_SS_SRBs_DRBs_Config(p_CellId, p_IOT_State, v_TimingInfo, omit);
  }
     


After Change:

	function f_NBIOT_RRC_ConnectionRelease_Local(NBIOT_CellId_Type p_CellId,
                                               IOT_STATE_Type p_IOT_State,
                                               SubFrameTiming_Type p_TimingAtT,
                                               boolean p_isDrbConfigured := true /*WA#22_4_15*/) runs on NBIOT_PTC
  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs (and default DRB)
     *       => this will in fact happen in the future */
    var SubFrameTiming_Type v_Timing := p_TimingAtT;
    var template (value) TimingInfo_Type v_TimingInfo;
    /* Step 1 at T: disable the UL grant */
    f_NBIOT_StopULGrantTransmission(p_CellId);
    /* Step 2 at T+64ms */
    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 64);
    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);
    if (p_IOT_State == USER_PLANE) {
      /* release security */
      f_NBIOT_SS_RRC_ReleaseSecurity(p_CellId, v_TimingInfo);
    }
    f_NBIOT_ULGrantTransmission(p_CellId, cs_NB_UL_GrantScheduling_Start(cs_NB_PeriodicGrant(-, 4)), v_TimingInfo);  /* @sic R5-177097: sic@
                                                                                                                        give 4 sub-sequent UL grants to allow the UE to do RLC ACK */
    /* Step 3 at T+80ms: Release SRBs and DRBs */
    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 80);
    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);
    f_NBIOT_SS_SRBs_DRBs_Release(p_CellId, p_IOT_State, v_TimingInfo, -, -, p_isDrbConfigured); //WA#22_4_15
    /* Step 4 at T+96ms: Configure SRBs and DRBs */
    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 96);
    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);
    f_NBIOT_SS_SRBs_DRBs_Config(p_CellId, p_IOT_State, v_TimingInfo, omit);
  }



MCC160 Implementation
	  function f_NBIOT_SS_ActiveDRBs(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return DRB_IdentityList_Type

  {

    var NB_PdcpCountInfoList_Type v_PdcpCountInfoList := f_NBIOT_SS_PdcpCount_GetAll(p_CellId);

    var DRB_IdentityList_Type v_DrbIdList := {};

    var integer k := 0;

    var integer i;

    for (i := 0; i < lengthof(v_PdcpCountInfoList); i := i + 1) {

      if (ischosen(v_PdcpCountInfoList[i].RadioBearerId.Drb)) {

        v_DrbIdList[k] := v_PdcpCountInfoList[i].RadioBearerId.Drb;

        k := k + 1;

      }

    }

    return v_DrbIdList;

  }

	  function f_NBIOT_RRC_ConnectionRelease_Local(NBIOT_CellId_Type p_CellId,

                                               IOT_STATE_Type p_IOT_State,

                                               SubFrameTiming_Type p_TimingAtT) runs on NBIOT_PTC

  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs (and default DRB)

     *       => this will in fact happen in the future */

    var SubFrameTiming_Type v_Timing := p_TimingAtT;

    var template (value) TimingInfo_Type v_TimingInfo;

    var DRB_IdentityList_Type v_DrbIdList := f_NBIOT_SS_ActiveDRBs(p_CellId);

    /* Step 1 at T: disable the UL grant */

    f_NBIOT_StopULGrantTransmission(p_CellId);

    /* Step 2 at T+64ms */

    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 64);

    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);

    if (p_IOT_State == USER_PLANE) {

      /* release security */

      f_NBIOT_SS_RRC_ReleaseSecurity(p_CellId, v_TimingInfo);

    }

    f_NBIOT_ULGrantTransmission(p_CellId, cs_NB_UL_GrantScheduling_Start(cs_NB_PeriodicGrant(-, 4)), v_TimingInfo);  /* @sic R5-177097: sic@

                                                                                                                        give 4 sub-sequent UL grants to allow the UE to do RLC ACK */

    /* Step 3 at T+80ms: Release SRBs and DRBs */

    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 80);

    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);

    f_NBIOT_SS_SRBs_DRBs_Release(p_CellId, p_IOT_State, v_TimingInfo, v_DrbIdList);
    /* Step 4 at T+96ms: Configure SRBs and DRBs */

    v_Timing := f_SubFrameTiming_AddMilliSeconds(p_TimingAtT, 96);

    v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_Timing);

    f_NBIOT_SS_SRBs_DRBs_Config(p_CellId, p_IOT_State, v_TimingInfo, omit);

  }

	  function f_NBIOT_SS_SRBs_DRBs_Reset(NBIOT_CellId_Type  p_CellId,

                                      IOT_STATE_Type p_IOT_State,

                                      DelayBeforeConfiguration_Type p_DelayBeforeReset := noDelay,

                                      boolean p_StopULGrant := false,

                                      template (omit) boolean p_CnfFlag := omit) runs on NBIOT_PTC

  { /* Note: this function shall use timing info "Now" only otherwise between release and config there have to be at least 5ms */

    var DRB_IdentityList_Type v_DrbIdList := f_NBIOT_SS_ActiveDRBs(p_CellId);   // @sic R5s180547 sic@
    f_DelayBeforeConfiguration(p_DelayBeforeReset);

    if (p_StopULGrant) {

      f_NBIOT_StopULGrantTransmission(p_CellId);

    }

    f_NBIOT_SS_SRBs_DRBs_Release(p_CellId, p_IOT_State, -, v_DrbIdList, p_CnfFlag);

    f_NBIOT_SS_SRBs_DRBs_Config(p_CellId, p_IOT_State, -, -, p_CnfFlag);

  }


1.5 f_NBIOT_SS_SRBs_DRBs_Release
	Template name
	f_NBIOT_SS_SRBs_DRBs_Release

	Reason for change
	As a result of change 1.3, this function also needs to be modified to accommodate the flag usage specified in change 1.1.


	Summary of change
	Modified TTCN code by introducing a flag to ensure the DRB on NCell 11 is released only once.

	TTCN module
	NBIOT_ConfigurationSteps.ttcn

	MCC160 Comment
	Not needed


Before Change:

	function f_NBIOT_SS_SRBs_DRBs_Release(NBIOT_CellId_Type p_CellId,
                                        IOT_STATE_Type p_IOT_State,
                                        template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                        DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,
                                        template (omit) boolean p_CnfFlag := omit) runs on NBIOT_PTC
  {
    var template (value) NB_RadioBearerList_Type v_RadioBearerList;
    var integer i;
    var integer k;
    v_RadioBearerList[0] := cs_NB_SRB_Release(tsc_SRB1bis);
    if (p_IOT_State == USER_PLANE) {
      v_RadioBearerList[1] := cs_NB_SRB_Release(tsc_SRB1);
      i := lengthof(v_RadioBearerList);
      for (k:=0; k < lengthof(p_DrbList); k:= k+1) {
        v_RadioBearerList[i+k] := cs_NB_DRB_Release(p_DrbList[k]);
      }
    }
    f_NBIOT_SS_CommonRadioBearerConfig(p_CellId, v_RadioBearerList, p_TimingInfo, p_CnfFlag);
  }
     


After Change:

	function f_NBIOT_SS_SRBs_DRBs_Release(NBIOT_CellId_Type p_CellId,
                                        IOT_STATE_Type p_IOT_State,
                                        template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                        DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,
                                        template (omit) boolean p_CnfFlag := omit,
                                        boolean p_DrbConfigured := true /*WA#22_4_15*/) runs on NBIOT_PTC
  {
    var template (value) NB_RadioBearerList_Type v_RadioBearerList;
    var integer i;
    var integer k;
    v_RadioBearerList[0] := cs_NB_SRB_Release(tsc_SRB1bis);
    if (p_IOT_State == USER_PLANE) {
      v_RadioBearerList[1] := cs_NB_SRB_Release(tsc_SRB1);
      if (p_DrbConfigured) {
      i := lengthof(v_RadioBearerList);
      for (k:=0; k < lengthof(p_DrbList); k:= k+1) {
        v_RadioBearerList[i+k] := cs_NB_DRB_Release(p_DrbList[k]);
      }
     }
    }
    f_NBIOT_SS_CommonRadioBearerConfig(p_CellId, v_RadioBearerList, p_TimingInfo, p_CnfFlag);
  }



